

6. INTEGRACIÓN. EJERCICIOS.

EJERCICIO 6.1. (a) Obténgase una primitiva de la función $f(x)=p(x)/q(x)$, donde $p(x)=3x^3+14x+15$ y $q(x)=x^4-8x^3+23x^2-28x+12$.

(b) Calcúlese la integral definida entre 0 y 4 de $f(x)=x^2e^x$.

SOLUCIÓN. (a) Se define la función f:

```
(%i1) define(p(x), 3*x^3+14*x+15); define(q(x), x^4-8*x^3+23*x^2-28*x+12);
      define(f(x), p(x)/q(x));
(%o1) p(x):= 3 x^3 + 14 x + 15
(%o2) q(x):= x^4 - 8 x^3 + 23 x^2 - 28 x + 12
(%o3) f(x):= (3 x^3 + 14 x + 15) / (x^4 - 8 x^3 + 23 x^2 - 28 x + 12)
```

Una primitiva de f viene dada por

```
(%i4) integrate(f(x), x);
(%o4) -16 log(x-1) - 50 log(x-2) + 69 log(x-3) + 67 / (x-2)
```

(b) Se define la función f:

```
(%i5) define(f(x), x^2*%e^x);
(%o5) f(x):= x^2 %e^x
```

La integral definida entre 0 y 4 de f es

```
(%i6) integrate(f(x), x, 0, 4);
(%o6) 10 %e^4 - 2
```

EJERCICIO 6.2. (a) Prográmesse una función que dé como salida la aproximación de la integral definida de una función $f:R \rightarrow R$ en el intervalo $[a,b]$ utilizando la fórmula compuesta del trapecio. Los parámetros de entrada deben ser la función f, los extremos a y b del intervalo y el número de subintervalos n utilizados para obtener la aproximación.

(b) Prográmesse una función con los mismos parámetros de entrada que la del apartado (a) que devuelva como salida la aproximación de la integral definida de una función $f:R \rightarrow R$ en el intervalo $[a,b]$ utilizando, en este caso, la regla de Simpson compuesta.

(c) Ejecútense las funciones de los apartados (a) y (b) para $f(x)=\sin(x^2+1)$ y el intervalo $[0,\pi]$, con $n=5$, $n=100$ y $n=200$, respectivamente.

SOLUCIÓN. (a) La función que se pide es la siguiente:

```
(%i7) Trapecios(f,a,b,n):=
      block(
        do
          (define(f(x),f),
            h: (b-a)/n,
            n_int: makelist(a+h*i,i,1,n-1,1),
            fn_int: makelist(f(n_int[i]),i,1,n-1,1),
            T: float(h/2*(f(a)+2*apply("+",fn_int)+f(b))),
            return(T)
          )
      )$
```

En Trapecios(f,a,b,n), n_int denota los nodos interiores al intervalo [a,b], es decir, los nodos que se utilizan para aproximar la integral menos el nodo inicial a y el nodo final b. Para la fórmula compuesta del trapecio los nodos son

$x_0 = a + h \cdot 0 = a$, $x_1 = a + h \cdot 1$, $x_2 = a + h \cdot 2, \dots, x_{(n-1)} = a + h \cdot (n-1)$, $x_n = a + h \cdot n = b$, donde $h = (b-a)/n$.

Luego los nodos interiores son

$x_1 = a + h \cdot 1$, $x_2 = a + h \cdot 2, \dots, x_{(n-1)} = a + h \cdot (n-1)$,

que se generan utilizando el comando lista, como sigue:

`n_int: makelist(a+h*i,i,1,n-1,1)`

A continuación, se evalúa la función f en los nodos interiores y se guardan los valores obtenidos en la lista fn_int:

`fn_int: makelist(f(n_int[i]),i,1,n-1,1).`

Por último, la aproximación por el método de los trapecios viene dada por T:

`T: float((b-a)/(2*n)*(f(a)+2*apply("+",fn_int)+f(b))),`

que es el valor que devuelve el programa.

(b) La función sigue el mismo esquema que la del apartado (a), utilizando ahora la regla de Simpson compuesta.

```
(%i8) Simpson(f,a,b,n):=
      block(
        do
          (define(f(x),f),
            h: (b-a)/n,
            n_int: makelist(a+h/2*i,i,1,2*n-1,1),
            fn_int_impares: makelist(f(n_int[i]),i,1,2*n-1,2),
            fn_int_pares: makelist(f(n_int[i]),i,2,2*n-1,2),
            S: float(h/6*(f(a)+4*apply("+",fn_int_impares)+2*apply("+",fn_int_pares)+f(b))),
            return(S)
          )
      )$
```

En este caso, al aplicar la fórmula de Simpson para n subintervalos, se utilizan $2n + 1$ nodos, que vienen dados por

$$x_0 = a + h/2 * 0, x_1 = a + h/2 * 1, x_2 = a + h/2 * 2, \dots, x_{(2n-1)} = a + h/2 * (2n-1), x_{(2n)} = a + h/2 * 2n = b.$$

Así, pues los nodos interiores son

$$x_1 = a + h/2 * 1, x_2 = a + h/2 * 2, \dots, x_{(2n-1)} = a + h/2 * (2n-1)$$

y se obtienen con el comando `makelist(a+h/2*i,i,1,2*n-1,1)`.

Por otra parte, las listas `fn_int_impares` y `fn_int_pares` contienen los valores de f en los nodos interiores impares y pares, respectivamente. Así, la regla de Simpson compuesta para n subintervalos se expresa del siguiente modo:

$$S: \text{float}((b-a)/(6*n)*(f(a)+4*\text{apply}("+",fn_int_impares)+2*\text{apply}("+",fn_int_pares)+f(b))).$$

(c) Para $f(x) = \sin(x^2 + 1)$ se tiene que

```
(%i9) define(f(x), sin(x^2+1));
      print("Utilizando la función Trapecios para n=5, n=20 y n=200:")$
      Trapecios(f(x),0,%pi,5); Trapecios(f(x),0,%pi,20); Trapecios(f(x),0,%pi,200);
      print("Utilizando la función Simpson para n=5, n=20 y n=200:")$
      Simpson(f(x),0,%pi,5); Simpson(f(x),0,%pi,20); Simpson(f(x),0,%pi,200);
      print("Utilizando el comando quad_qags:")$ quad_qags(f(x), x, 0, %pi);
```

```
(%o9) f(x):= sin(x^2 + 1)
```

Utilizando la función Trapecios para n=5, n=20 y n=200:

```
(%o11) 0.82573853780894
```

```
(%o12) 0.89179589264346
```

```
(%o13) 0.89346394279539
```

Utilizando la función Simpson para n=5, n=20 y n=200:

```
(%o15) 0.90585532791374
```

```
(%o16) 0.89349566732159
```

```
(%o17) 0.89348018144381
```

Utilizando el comando quad_qags:

```
(%o19) [0.89348017999346, 7.0902268146960138 10^-13, 63, 0]
```

El comando `quad_qags` proporciona una aproximación de la integral de f en el intervalo $[0, \pi]$ con un error estimado igual a $7.0902268146960138 \cdot 10^{-13}$, por lo que es muy buena aproximación.

Obsérvese que para $n=5$, el valor que proporciona la regla de Simpson compuesta se aproxima mucho más al que resulta de ejecutar `quad_qags` que el que viene dado por la fórmula compuesta del trapecio.

Como es lógico, al aumentar el número de subdivisiones en $n=20$ y $n=200$ se obtienen mejores aproximaciones.

EJERCICIO 6.3. Calcúlese la integral doble de la función $f(x,y) = yx^x$ sobre el rectángulo $[0,1] \times [0,2]$.

Se define primero la función f :

```
(%i20) define(f(x,y), y*x*%e^x);
(%o20) f(x,y) := x %e^x y
```

La integral que se pide es (se integra primero respecto a la variable y, y después respecto a la variable x)

```
(%i21) integrate(integrate(f(x,y),y,0,2),x,0,1);
(%o21) 2
```

Como f es continua, por el Teorema de Fubini se puede cambiar el orden de integración (primero con respecto a la variable x, y después con respecto a la variable y):

```
(%i22) integrate(integrate(f(x,y),x,0,1),y,0,2);
(%o22) 2
```

EJERCICIO 6.4. Calcúlese la integral doble de la función $f(x,y)=x^2y$ sobre el recinto limitado por las gráficas de las funciones $g(x)=x^2-1$ y $h(x)=-x^2+2$.


SOLUCIÓN. Se definen f(x,y), g(x) y h(x):

```
(%i23) define(f(x,y), x^2*y); define(g(x),x^2-1); define(h(x),-x^2+2);
(%o23) f(x,y) := x^2 y
(%o24) g(x) := x^2 - 1
(%o25) h(x) := 2 - x^2
```

El recinto de integración es el siguiente:

```
(%i26) load(draw);
Loading C:/Users/user/maxima/binary/binary-gcl/share/draw/grcommon.o
Finished loading C:/Users/user/maxima/binary/binary-gcl/share/draw/grcommon.o
Loading C:/Users/user/maxima/binary/binary-gcl/share/draw/gnuplot.o
Finished loading C:/Users/user/maxima/binary/binary-gcl/share/draw/gnuplot.o
Loading C:/Users/user/maxima/binary/binary-gcl/share/draw/vtk.o
Finished loading C:/Users/user/maxima/binary/binary-gcl/share/draw/vtk.o
Loading C:/Users/user/maxima/binary/binary-gcl/share/draw/picture.o
Finished loading C:/Users/user/maxima/binary/binary-gcl/share/draw/picture.o
(%o26) C:/PROGRA~1/MAXIMA~1.0/share/maxima/5.30.0/share/draw/draw.lisp
```

```
(%i27) wxdraw2d(x_voxel = 30,y_voxel = 30,explicit(g(x),x,-2,2), explicit(h(x),x,-2,2),
region(y-g(x)>=0 and y-h(x)<=0,x, -2, 2,y,-2,2));
```



Los puntos de corte de las gráficas de g y h son:

```
(%i28) solve(g(x)-h(x));
```

```
(%o28) [x = -sqrt(3)/sqrt(2), x = sqrt(3)/sqrt(2)]
```

Así pues, la integral es igual a

```
(%i29) integrate(integrate(f(x,y),y,x^2-1,-x^2+2),x,-sqrt(3)/sqrt(2),sqrt(3)/sqrt(2));
```

```
(%o29) 3^(3/2)/(5*sqrt(2))
```

EJERCICIO 6.5. Calcúlese el volumen del sólido de revolución generado al girar la gráfica de la función $f(x)=\sin(x)+2$, con x variando en el intervalo $[0, 2\pi]$.

SOLUCIÓN. Se define f:


```
(%i30) define(f(x), sin(x)+2);
```

```
(%o30) f(x):=sin(x)+2
```

El sólido de revolución que se genera tiene ecuación implícita $f(x)^2-y^2-z^2=0$. Se representa en la siguiente figura:

(%i31) `wxdraw3d(implicit((f(x))^2-y^2-z^2=0, x,0,2*%pi,y,-3,3,z,-3,3), proportional_axes='xyz);`

(%t31)



(%o31)

Y el volumen del sólido de revolución viene dado por la integral definida:

(%i32) `%pi*integrate(f(x)^2,x,0,2*%pi);`

(%o32) $9\pi^2$